query: embed.html##title: Carte du métro de Xi'an
  #defaultView:Map{"hide":["?coords", "?line","?rgb"]}
  SELECT ?station ?stationLabel ?subwayLine ?coords ?line ?layer ?rgb WHERE {
    VALUES ?search {
      wd:Q555733#critère de recherche de système de métro
    }
        ?search wdt:P527 ?lignes.#quelles sont les lignes de ce métro
        ?lignes wdt:P559 ?termini.#quels sont les terminus de ce métro
    ?station wdt:P31 wd:Q928830;  #station de métro
             wdt:P361 ?search;#partie du réseau recherché
      wdt:P625 ?coords;
      wdt:P81 ?subwayLine;
      wdt:P197 ?pred.
    ?pred wdt:P625 ?coords_pred;
      wdt:P81 ?subwayLine_pred.
    ?station p:P197 _:b1.
    _:b1 ps:P197 ?pred ;
      pq:P5051 ?towards;
      pq:P81 ?line_pq.
    FILTER(?subwayLine_pred = ?lignes)#on ne prend les LINES que si la correspondance est sur la même ligne
    FILTER(?subwayLine = ?lignes)
    FILTER(?towards = ?termini) 
    FILTER(?station != ?termini)
    MINUS { ?station (wdt:P576|wdt:P582|wdt:P3999) ?dispar. }
    MINUS {
      ?station wdt:P5817 ?interdit.
      VALUES ?interdit {
        wd:Q811683
        wd:Q63065035 
        wd:Q12377751
        wd:Q55653430
        wd:Q30108381
        wd:Q55570340
        wd:Q11639308
      }
    }
    ?pred (p:P625/psv:P625/wikibase:geoLatitude) ?lat1   ;
      (p:P625/psv:P625/wikibase:geoLongitude) ?lon1  .
    ?station (p:P625/psv:P625/wikibase:geoLatitude) ?lat2 ;
      (p:P625/psv:P625/wikibase:geoLongitude) ?lon2.
    BIND(CONCAT("LINESTRING(", STR(?lon1), " ", STR(?lat1),  ",", STR(?lon2), " ", STR(?lat2), ")") AS ?str)
    BIND(STRDT(?str, geo:wktLiteral) AS ?line  )
    ?subwayLine wdt:P465 ?rgb   .
    SERVICE wikibase:label {
      bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en" .
      ?subwayLine rdfs:label ?layer .
      ?pred rdfs:label ?predLabel .
      ?station rdfs:label ?stationLabel .
    } 
  }
  
  GROUP BY ?station ?stationLabel  ?subwayLine ?rgb ?predLabel ?layer  ?coords ?line